NAG Library Routine Document

G02AAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G02AAF computes the nearest correlation matrix, in the Frobenius norm, to a given square, input matrix.

2 Specification

```
SUBROUTINE G02AAF(G, LDG, N, ERRTOL, MAXITS, MAXIT, X, LDX, ITER, FEVAL,
1                 NRMGRD, IFAIL)
INTEGER           LDG, N, MAXITS, MAXIT, LDX, ITER, FEVAL, IFAIL
double precision  G(LDG,N), ERRTOL, X(LDX,N), NRMGRD
```

3 Description

A correlation matrix may be characterized as a real square matrix that is symmetric, has a unit diagonal and is positive semi-definite.

G02AAF applies an inexact Newton method to a dual formulation of the problem, as described by Qi and Sun (2006). It applies the improvements suggested by Borsdorf (2007).

4 References

Borsdorf R (2007) A Newton Algorithm for the Nearest Correlation Matrix. *M Sc Dissertation* School of Mathematics, University of Manchester

Qi H and Sun D (2006) A Quadratically Convergent Newton Method for Computing the Nearest Correlation Matrix *SIAM J. Matrix AnalAppl* **29(2)** 360–385

5 Parameters

1: G(LDG,N) – ***double precision*** array *Input/Output*

On entry: G, the initial matrix.

On exit: a symmetric matrix $\frac{1}{2}\left(G+G^{\mathrm{T}}\right)$ with the diagonal set to I.

2: LDG – INTEGER *Input*

On entry: the first dimension of the array G as declared in the (sub)program from which G02AAF is called.

Constraint: LDG $\geq$ N.

3: N – INTEGER *Input*

On entry: the size of the matrix G.

Constraint: N $>$ 0.

4: ERRTOL – ***double precision*** *Input*

On entry: the termination tolerance for the Newton iteration. If ERRTOL $\leq$ 0.0 then $n \times \sqrt{\textbf{\textit{machine precision}}}$ is used.

5: MAXITS – INTEGER *Input*

On entry: MAXITS specifies the maximum number of iterations used for the iterative scheme used to solve the linear algebraic equations at each Newton step.

If MAXITS ≤ 0, $2 \times$ N is used.

6: MAXIT – INTEGER *Input*

On entry: specifies the maximum number of Newton iterations.

If MAXIT ≤ 0, 200 is used.

7: X(LDX,N) – ***double precision*** array *Output*

On exit: contains the nearest correlation matrix.

8: LDX – INTEGER *Input*

On entry: the first dimension of the array X as declared in the (sub)program from which G02AAF is called.

Constraint: LDX $\geq$ N.

9: ITER – INTEGER *Output*

On exit: the number of Newton steps taken.

10: FEVAL – INTEGER *Output*

On exit: the number of function evaluations of the dual problem.

11: NRMGRD – ***double precision*** *Output*

On exit: the norm of the gradient of the last Newton step.

12: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, N ≤ 0,
or LDG $<$ N,
or LDX $<$ N.

IFAIL $= 2$

The routine fails to converge in MAXIT iterations. Increase MAXIT or check the call to the routine.

IFAIL = 3

Machine precision is limiting convergence. In this instance the returned value of X may be useful.

IFAIL = 4

An internal eigenproblem could not be solved. This should not occur. Please contact NAG with details of your call.

IFAIL = −999

Internal memory allocation failed.

7 Accuracy

The returned accuracy is controlled by ERRTOL and limited by ***machine precision***.

8 Further Comments

None.

9 Example

This example finds the nearest correlation matrix to:

$$G = \begin{pmatrix} 2 & -1 & 0 & 0 \\ -1 & 2 & -1 & 0 \\ 0 & -1 & 2 & -1 \\ 0 & 0 & -1 & 2 \end{pmatrix}$$

9.1 Program Text

```
*     G02AAF Example Program Text
*     Mark 22 Release. NAG Copyright 2008.
*     .. Parameters ..
      INTEGER          NMAX, LDG, LDX, NOUT
      PARAMETER        (NMAX=4,LDG=5,LDX=NMAX,NOUT=6)
*     .. Local Scalars ..
      DOUBLE PRECISION ERRTOL, NRMGRD
      INTEGER          FEVAL, I, IFAIL, ITER, J, MAXIT, MAXITS, N
*     .. Local Arrays ..
      DOUBLE PRECISION G(LDG,NMAX), X(LDX,NMAX)
*     .. External Subroutines ..
      EXTERNAL         G02AAF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G02AAF Example Program Results'
      WRITE (NOUT,*)
*
*     Set up matrix G
      N = 4
      DO 40 J = 1, N
         DO 20 I = 1, N
            G(I,J) = 0.0
   20    CONTINUE
         G(J,J) = 2.0D0
   40 CONTINUE
      DO 60 J = 2, N
         G(J-1,J) = -1.0D0
         G(J,J-1) = -1.0D0
   60 CONTINUE
*
*     Set up method parameters
      ERRTOL = 1.0D-7
      MAXITS = 200
      MAXIT = 10
*
```

```
      IFAIL = 1
      CALL G02AAF(G,LDG,N,ERRTOL,MAXITS,MAXIT,X,LDX,ITER,FEVAL,NRMGRD,
     +            IFAIL)
*
      IF (IFAIL.EQ.0) THEN
         WRITE (NOUT,*) '     Nearest Correlation Matrix'
         WRITE (NOUT,99999) ((X(I,J),J=1,NMAX),I=1,NMAX)
         WRITE (NOUT,*)
         WRITE (NOUT,99998) ' Number of Newton steps taken:', ITER
         WRITE (NOUT,99997) ' Number of function evaluations:', FEVAL
         IF (NRMGRD.GT.ERRTOL) THEN
            WRITE (NOUT,99996) ' Norm of gradient of last Newton step:',
     +        NRMGRD
         END IF
      ELSE
         WRITE (NOUT,99995) IFAIL
      END IF
      WRITE (NOUT,*)
*
99999 FORMAT (2X,4F11.5)
99998 FORMAT (A,I11)
99997 FORMAT (A,I9)
99996 FORMAT (A,E11.3)
99995 FORMAT (' ** G02AAF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 G02AAF Example Program Results

      Nearest Correlation Matrix
      1.00000   -0.80841    0.19159    0.10678
     -0.80841    1.00000   -0.65623    0.19159
      0.19159   -0.65623    1.00000   -0.80841
      0.10678    0.19159   -0.80841    1.00000

 Number of Newton steps taken:          3
 Number of function evaluations:        4
```